# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=oniongrok
pkgver=v1.1.11.r0.gcbae9e7
pkgrel=1
pkgdesc='Onion addresses for anything.'
arch=('x86_64' 'aarch64')
groups=('blackarch' 'blackarch-tunnel')
url='https://github.com/cmars/oniongrok'
license=('MIT')
depends=('glibc')
makedepends=('git' 'go')
source=("git+https://github.com/cmars/$pkgname.git")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname

  git describe --long --tags | sed 's/\([^-]*-g\)/r\1/;s/-/./g'
}

build() {
  cd $pkgname

  GOPATH="$srcdir" go mod download
  GOPATH="$srcdir" go build \
    -trimpath \
    -buildmode=pie \
    -mod=readonly \
    -modcacherw \
    -ldflags "-s -w" \
    -o $pkgname .

}

package() {
  cd $pkgname

  install -Dm 755 $pkgname "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md ATTRIBUTIONS

  install -Dm 644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

